
Sesión 20 (A) — El contraste F con datos reales: tres restricciones
sobre hprice2

Marcos Bujosa

Descripción de la práctica
La lección 13 construyó el contraste F comparando dos ajustes anidados: el del modelo completo y el del

modelo que respeta H0. La diferencia SRCr − SRC es lo que la restricción cuesta en longitud al cuadrado;
dividida por el número de restricciones q y por SRC/(n − k) da el F . Hoy hacemos esa cuenta con datos
reales, restricción por restricción, y después dejamos que Gretl la haga con restrict.

Retomamos hprice2 con el modelo de la lección 13, price ~ rooms + nox + stratio, y sus tres res-
tricciones: excluir nox y stratio a la vez, excluir solo stratio, e imponer que los coeficientes de nox y
stratio sean iguales. La primera es un F con q = 2; la segunda, un F con q = 1, que debe coincidir con el
cuadrado de un t; la tercera no es una exclusión, y obliga a construir el modelo restringido a mano.

Objetivos

1. Estimar el modelo completo y el restringido, y construir el F con las dos SRC, q y n − k.

2. Comprobar que restrict devuelve el mismo F y leer su salida: el estadístico, el valor p y las estima-
ciones restringidas.

3. Ver que con q = 1 el F es el cuadrado del t, y el valor crítico Fc el cuadrado de tc.

4. Imponer una igualdad entre dos coeficientes construyendo la suma de los regresores como regresor
único.

5. Reencontrar el caso nox y lnox de la lección 13, que la práctica B de la sesión 22 (actividad 1 (html))
explicará.

6. Dibujar la elipse de confianza de los coeficientes de nox y lnox y compararla con el rectángulo de
sus dos intervalos: el intervalo de confianza en dos dimensiones y el efecto de la correlación entre los
estimadores.

Comandos nuevos de esta práctica. Seguimos usando ols con la opción --quiet, los accesores $ess,
$coeff y $stderr (sesiones 11 a 18), las funciones critical y pvalue y el bloque restrict ... end
restrict (práctica A de la sesión 18, actividad 3 (html)). Lo nuevo:

b[nox] dentro de restrict, un coeficiente puede nombrarse por su posición en la lista de regresores, b[3],
como en la sesión 18, o por el nombre de su variable, b[nox]. Cada línea del bloque es una restricción:
una combinación lineal de coeficientes a la izquierda del igual y un número a la derecha.

critical(F, q, gl, 0.05) el valor de la F de Fisher con q y gl grados de libertad que deja una proba-
bilidad 0,05 a su derecha; pvalue(F, q, gl, x) es la probabilidad de que esa F supere x. Son las
funciones de la sesión 18 con la letra F en lugar de t, y con dos grados de libertad en lugar de uno.

%.4e un formato más de printf: notación científica con cuatro decimales, cómoda para las sumas de cua-
drados. Los otros dos que se usan, %d (un número entero) y %.3g (tres cifras significativas, en notación
decimal o científica según su tamaño), ya aparecieron en las sesiones 7, 11, 16 y 18.

Licencia: Creative Commons Attribution-ShareAlike 4.0 International (CC BY-SA 4.0).

$vcv la matriz de varianzas y covarianzas de los coeficientes del último modelo. Entre corchetes, $vcv[3:4,3:4],
se toma el bloque de las filas y columnas 3 y 4: las varianzas de los coeficientes tercero y cuarto en la
diagonal y su covarianza fuera de ella. $coeff[3:4] toma del mismo modo las dos estimaciones. Están
en la Actividad 6.

matrix, cholesky, seq, ones, ~ en la Actividad 6, para recorrer la elipse punto a punto: matrix declara
una matriz; cholesky(V) devuelve la matriz triangular L con L*L' = V; seq(0, 360) es la fila de los
enteros de 0 a 360; ones(r, 1) una columna de r unos; ~ pega dos matrices lado a lado; y el apóstrofo
traspone.

plot con una matriz el bloque plot ... end plot de la sesión 11 admite una matriz en lugar de una
lista de series: dibuja sus columnas, la última en el eje horizontal. set force_decpoint on hace que
sprintf escriba los decimales con punto, como exige gnuplot, y @rect dentro de una línea literal
inserta el contenido de la cadena rect.

Actividad 1 - El modelo completo
Abra hprice2 y estime price sobre rooms, nox y stratio: Modelo ->Mínimos cuadrados ordinarios.

Anote la suma de cuadrados de los residuos y los grados de libertad, n − k = 506 − 4 = 502.
El bloque abre hprice2, estima el modelo completo con ols y guarda en escalares lo que hará falta en

toda la práctica: el número de observaciones n, el número de coeficientes k (la constante y tres regresores),
la suma de cuadrados de los residuos SRC, que Gretl devuelve en el accesor $ess, y el denominador del F ,
s2 = SRC/(n-k), que es s2.

o bien teclee en línea de comandos:

open hprice2.gdt --quiet

ols price const rooms nox stratio
scalar n = $nobs
scalar k = 4
scalar SRC = $ess
scalar s2 = SRC/(n-k)
printf "\nSRC = %.4e n-k = %d s^2 = SRC/(n-k) = %.4e\n", SRC, n-k, s2

Modelo 1: MCO, usando las observaciones 1-506
Variable dependiente: price

coeficiente Desv. típica Estadístico t valor p
----------------------------------------------------------------
const 9458,01 4371,41 2,164 0,0310 **
rooms 7000,63 409,150 17,11 4,83e-52 ***
nox -1693,33 236,250 -7,168 2,75e-12 ***
stratio -1166,96 128,764 -9,063 2,85e-18 ***

Media de la vble. dep. 22511,51 D.T. de la vble. dep. 9208,856
Suma de cuad. residuos 1,71e+10 D.T. de la regresión 5837,767
R-cuadrado 0,600521 R-cuadrado corregido 0,598133
F(3, 502) 251,5451 Valor p (de F) 1,31e-99
Log-verosimilitud -5104,059 Criterio de Akaike 10216,12
Criterio de Schwarz 10233,02 Crit. de Hannan-Quinn 10222,75

SRC = 1,7108e+10 n-k = 502 s^2 = SRC/(n-k) = 3,4080e+07

Lo que debe observar

Los tres regresores tienen t grandes; el resultado de las tres restricciones de hoy es previsible, y esa es la
intención: lo que interesa es la mecánica. El denominador del F , SRC/(n − k), es s2, el cuadrado de la “D.T.
de la regresión” de la tabla. Compruébelo: 5837,772 = 3,408 · 107.

Actividad 2 - Excluir dos regresores: el F a mano y con restrict
La hipótesis H0 : βnox = βstratio = 0 dice que el modelo es price ~ rooms. Estime ese modelo restringido,

anote su SRCr y construya el F con q = 2.
Estimamos el modelo restringido con ols y --quiet, que evita imprimir la tabla: solo queremos su $ess,

que guardamos en SRCr. Con q = 2 restricciones, F es la fórmula de la lección 13 escrita tal cual: el aumento
de la SRC por restricción, dividida por s2. Cuatro printf: el primero imprime SRCr; el segundo, el aumento
de la SRC; el tercero, el F ; y el cuarto, el valor crítico de la F con 2 y n − k grados de libertad al 5 % y el
valor p.

en línea de comandos:

ols price const rooms --quiet
scalar SRCr = $ess
scalar q = 2
scalar F = ((SRCr - SRC)/q)/s2
printf "Modelo restringido (price ~ rooms): SRC_r = %.4e\n", SRCr
printf "Aumento de la SRC: SRC_r - SRC = %.4e\n", SRCr - SRC
printf "F = ((SRC_r - SRC)/%d)/(SRC/(n-k)) = %.3f\n", q, F
printf "F_c(%d,%d) al 5%% = %.3f valor p = %.3g\n", q, n-k, critical(F, q, n-k, 0.05), pvalue(F, q, n-k, F)

Modelo restringido (price ~ rooms): SRC_r = 2,2093e+10
Aumento de la SRC: SRC_r - SRC = 4,9854e+09
F = ((SRC_r - SRC)/2)/(SRC/(n-k)) = 73,143
F_c(2,502) al 5% = 3,014 valor p = 1,33e-28

Ahora pida a Gretl el mismo contraste. Vuelva a estimar el modelo completo y, en la ventana del modelo,
Contrastes ->Restricciones lineales; escriba las dos restricciones b[nox] = 0 y b[stratio] = 0, una
por línea.

Primero volvemos a estimar el modelo completo, porque restrict actúa siempre sobre el último modelo
estimado, y el último fue el restringido. El bloque restrict ... end restrict lleva una restricción por
línea, con los coeficientes nombrados por su variable. Al cerrar el bloque, Gretl imprime el F , su valor p y
las estimaciones del modelo que respeta las restricciones.

o bien teclee en línea de comandos:

ols price const rooms nox stratio --quiet
restrict

b[nox] = 0
b[stratio] = 0

end restrict

Conjunto de restricciones
1: b[nox] = 0
2: b[stratio] = 0

Estadístico de contraste: F(2, 502) = 73,1429, con valor p = 1,32811e-28

Estimaciones restringidas:

coeficiente Desv. típica Estadístico t valor p
--------------------------------------------------------------
const -34796,2 2651,53 -13,12 4,74e-34 ***
rooms 9119,55 419,339 21,75 1,87e-74 ***

nox 0,00000 0,00000 NA NA
stratio 0,00000 0,00000 NA NA

Desviación típica de la regresión = 6620,87

Lo que debe observar

El F de restrict es el que usted acaba de construir con las dos SRC; la coincidencia es exacta porque
son la misma cuenta. La salida añade las estimaciones restringidas: los coeficientes del modelo que respeta
H0. El de rooms no es el del modelo completo. Al quitar nox y stratio, la proyección de price cae sobre
un subespacio más pequeño y la coordenada de rooms se recalcula: es la figura de la proyección en dos pasos
de la lección 13, con ŷr en el plano de 1 y rooms.

La “D.T. de la regresión” del modelo restringido (la s de la lección 11), 6620,9, es mayor que la del
completo, 5837,8. Siempre lo es: el residuo restringido es más largo. Que la diferencia sea grande o pequeña
respecto al ruido es lo que el F juzga.

Actividad 3 - Excluir un solo regresor: F = t2

Con q = 1 la restricción βstratio = 0 se puede contrastar de dos maneras: con el F de los dos ajus-
tes anidados o con el t de la tabla del modelo completo. La lección 13 dijo que son el mismo contraste.
Compruébelo.

Modelo restringido sin stratio, con su $ess guardado en SRCr1, y el F con q = 1. Después se vuelve a
estimar el modelo completo para leer el t de stratio: el cociente entre su coeficiente, $coeff(stratio), y
su error estándar, $stderr(stratio). Cuatro printf: SRCr del modelo sin stratio; el F ; el t de stratio
y su cuadrado; y el valor crítico de la F junto al cuadrado del de la t.

en línea de comandos:

ols price const rooms nox --quiet
scalar SRCr1 = $ess
scalar F1 = ((SRCr1 - SRC)/1)/s2
ols price const rooms nox stratio --quiet
scalar t_st = $coeff(stratio)/$stderr(stratio)
printf "Modelo restringido (price ~ rooms + nox): SRC_r = %.4e\n", SRCr1
printf "F = (SRC_r - SRC)/(SRC/(n-k)) = %.3f\n", F1
printf "t de stratio en el modelo completo = %.3f ; t^2 = %.3f\n", t_st, t_st^2
printf "F_c(1,%d) al 5%% = %.3f ; t_c^2 = %.3f\n", n-k, critical(F, 1, n-k, 0.05), critical(t, n-k, 0.025)^2

Modelo restringido (price ~ rooms + nox): SRC_r = 1,9907e+10
F = (SRC_r - SRC)/(SRC/(n-k)) = 82,133
t de stratio en el modelo completo = -9,063 ; t^2 = 82,133
F_c(1,502) al 5% = 3,860 ; t_c^2 = 3,860

Lo que debe observar

F = t2 con todos los decimales, y Fc(1, 502) = t2
c : 3,860 = 1,9652. El t bilateral al 5 % y el F con q = 1

al 5 % rechazan exactamente en los mismos casos. La diferencia es de lectura: el t tiene signo y dice en qué
dirección se aleja el coeficiente de cero; el F solo mide cuánto sube la SRC, y por eso sirve también cuando
las restricciones son varias.

Actividad 4 - Una igualdad entre coeficientes
La hipótesis H0 : βnox = βstratio no excluye ningún regresor, así que el modelo restringido no se obtiene

borrando una variable. Si los dos coeficientes son iguales a un mismo β, entonces βnox nox + βstratio stratio =
β (nox + stratio): el modelo restringido tiene como regresor la suma nox + stratio. Constrúyala (Añadir
->Definir nueva variable), estime el modelo con rooms y la suma, y calcule el F con q = 1.

Creamos con series el regresor suma, estimamos el modelo restringido con rooms y esa suma, y cons-
truimos el F con q = 1 como en la Actividad 3. Tres printf: SRCr; el aumento de la SRC; y el F con su
valor crítico y su valor p.

en línea de comandos:

series nox_mas_stratio = nox + stratio
ols price const rooms nox_mas_stratio --quiet
scalar SRCr2 = $ess
scalar F2 = ((SRCr2 - SRC)/1)/s2
printf "Modelo restringido (price ~ rooms + (nox + stratio)): SRC_r = %.4e\n", SRCr2
printf "Aumento de la SRC = %.4e\n", SRCr2 - SRC
printf "F = %.3f F_c(1,%d) al 5%% = %.3f valor p = %.4f\n", F2, n-k, critical(F, 1, n-k, 0.05), pvalue(F, 1, n-k, F2)

Modelo restringido (price ~ rooms + (nox + stratio)): SRC_r = 1,7229e+10
Aumento de la SRC = 1,2131e+08
F = 3,560 F_c(1,502) al 5% = 3,860 valor p = 0,0598

Y de nuevo con restrict, escribiendo la restricción como b[nox] - b[stratio] = 0.
De nuevo el modelo completo, y un bloque restrict con una sola línea: la diferencia de los dos coeficientes

igual a cero. La salida da el F y las estimaciones restringidas, en las que nox y stratio comparten coeficiente.
en línea de comandos:

ols price const rooms nox stratio --quiet
restrict

b[nox] - b[stratio] = 0
end restrict

Restricción:
b[nox] - b[stratio] = 0

Estadístico de contraste: F(1, 502) = 3,55973, con valor p = 0,059774

Estimaciones restringidas:

coeficiente Desv. típica Estadístico t valor p
----------------------------------------------------------------
const 9323,41 4381,94 2,128 0,0338 **
rooms 7055,85 409,139 17,25 1,06e-52 ***
nox -1297,48 108,880 -11,92 5,20e-29 ***
stratio -1297,48 108,880 -11,92 5,20e-29 ***

Desviación típica de la regresión = 5852,6

Lo que debe observar

La SRC sube poco: 1,21 · 108 sobre 1,71 · 1010, y el F vale 3,56, por debajo de Fc = 3,86. Al 5 % la
igualdad no se rechaza, con un valor p de 0,060: cerca del umbral, y la lección 12 ya dijo qué hacer con un
valor p, que es leerlo, no solo compararlo.

Las estimaciones restringidas de restrict muestran el mismo coeficiente para nox y stratio, −1297,5,
que es el coeficiente de la suma en su modelo a mano: son el mismo ajuste. Y conviene decir qué significa esta
igualdad: nox se mide en partes por cien millones y stratio en alumnos por profesor, así que “una unidad
de cada una cuesta lo mismo” es una coincidencia de escalas, no una hipótesis económica. La restricción
es un ejercicio de mecánica; en la lección 15 veremos restricciones con contenido, como la de rendimientos
constantes a escala.

Actividad 5 - Apertura: el caso nox y lnox
Sustituya stratio por lnox, el logaritmo de nox, y compare el F de excluir las dos medidas de conta-

minación con los t de cada una.
El modelo con lnox en lugar de stratio, esta vez sin --quiet para ver la tabla con los t, seguido de un

bloque restrict que excluye las dos medidas de contaminación a la vez.
en línea de comandos:

ols price const rooms nox lnox
restrict

b[nox] = 0
b[lnox] = 0

end restrict

Modelo 2: MCO, usando las observaciones 1-506
Variable dependiente: price

coeficiente Desv. típica Estadístico t valor p
----------------------------------------------------------------
const -8566,88 9983,25 -0,8581 0,3912
rooms 8161,84 418,340 19,51 1,65e-63 ***
nox 404,190 2198,85 0,1838 0,8542
lnox -13262,2 12655,7 -1,048 0,2952

Media de la vble. dep. 22511,51 D.T. de la vble. dep. 9208,856
Suma de cuad. residuos 1,99e+10 D.T. de la regresión 6290,373
R-cuadrado 0,536175 R-cuadrado corregido 0,533404
F(3, 502) 193,4352 Valor p (de F) 2,36e-83
Log-verosimilitud -5141,843 Criterio de Akaike 10291,69
Criterio de Schwarz 10308,59 Crit. de Hannan-Quinn 10298,32

Sin considerar la constante, el valor p más alto fue el de la variable 3 (nox)

Conjunto de restricciones
1: b[nox] = 0
2: b[lnox] = 0

Estadístico de contraste: F(2, 502) = 28,1754, con valor p = 2,52888e-12

Estimaciones restringidas:

coeficiente Desv. típica Estadístico t valor p
--------------------------------------------------------------
const -34796,2 2651,53 -13,12 4,74e-34 ***
rooms 9119,55 419,339 21,75 1,87e-74 ***
nox 0,00000 0,00000 NA NA
lnox 0,00000 0,00000 NA NA

Desviación típica de la regresión = 6620,87

Lo que debe observar

Ningún t llega a 1,05 en valor absoluto; el F de excluir las dos a la vez vale 28. Uno a uno, ninguno de
los dos regresores parece necesario; juntos, el contraste rechaza con claridad que sobren los dos. No es una
contradicción del contraste: es que las dos preguntas son distintas, y la lección 14 explicará por qué esta
pareja las separa tanto. La práctica B de la sesión 22 (actividad 1 (html)) le pondrá números.

Actividad 6 - La elipse de confianza de nox y lnox
Los dos intervalos al 95 % de la Actividad 5 contienen el cero y el F rechaza el par (0, 0). Dibuje la región

de pares (βnox, βlnox) que el F no rechaza al 5 % y compárela con el rectángulo de los dos intervalos.

La región que el F no rechaza es una elipse centrada en las estimaciones: el intervalo de confianza en dos
dimensiones. Gretl la dibuja desde la ventana del modelo, Análisis ->Elipse de confianza, eligiendo
los dos coeficientes. En línea de comandos hay que construirla, y para eso hacen falta las varianzas y la
covarianza de los dos estimadores.

El bloque Elipse estima de nuevo el modelo con --quiet, guarda tc y el valor crítico Fc de la F con 2 y
n−k grados de libertad, y toma de la tabla de coeficientes y de $vcv (la matriz de varianzas y covarianzas de
los coeficientes) las filas y columnas 3 y 4, las de nox y lnox: b es la pareja de estimaciones y V su matriz de
varianzas, 2 × 2. corrEst es la correlación entre los dos estimadores, la covarianza dividida por el producto
de las dos desviaciones típicas. Los cuatro escalares iL, iR, jL y jR son los extremos de los dos intervalos.
Después, restrict --quiet contrasta el par (0, 0) (imprime el estadístico, no las estimaciones restringidas)
y $test recoge su F . Tres printf: los intervalos, la correlación, y el F del par (0, 0) con su valor crítico y
su valor p.

o bien teclee en línea de comandos:

ols price const rooms nox lnox --quiet
scalar tc = critical(t, $df, 0.025)
scalar Fc = critical(F, 2, $df, 0.05)
scalar gl = $df
matrix b = $coeff[3:4]
matrix V = $vcv[3:4,3:4]
scalar corrEst = V[1,2]/sqrt(V[1,1]*V[2,2])
scalar iL = b[1] - tc*sqrt(V[1,1])
scalar iR = b[1] + tc*sqrt(V[1,1])
scalar jL = b[2] - tc*sqrt(V[2,2])
scalar jR = b[2] + tc*sqrt(V[2,2])
restrict --quiet

b[nox] = 0
b[lnox] = 0

end restrict
printf "IC 95%% nox: [%.1f ; %.1f] IC 95%% lnox: [%.1f ; %.1f]\n", iL, iR, jL, jR
printf "Correlacion entre los estimadores de nox y lnox: %.3f\n", corrEst
printf "El par (0,0): F = %.3f F_c(2,%d) = %.3f valor p = %.3g\n", $test, gl, Fc, $pvalue

Conjunto de restricciones
1: b[nox] = 0
2: b[lnox] = 0

Estadístico de contraste: F(2, 502) = 28,1754, con valor p = 2,52888e-12

IC 95% nox: [-3915,9 ; 4724,3] IC 95% lnox: [-38126,8 ; 11602,3]
Correlacion entre los estimadores de nox y lnox: -0,993
El par (0,0): F = 28,175 F_c(2,502) = 3,014 valor p = 2,53e-12

La elipse es el conjunto de pares c con (c − b)⊤V −1(c − b) ≤ 2Fc: el F de contrastar βnox = c1, βlnox =
c2 es el lado izquierdo dividido por 2, y no se rechaza mientras no supere Fc. El bloque ElipseFigura
la recorre punto a punto: cholesky(V) devuelve una matriz triangular L con LL⊤ = V , y los puntos
b +

√
2Fc L (cos θ, sin θ)⊤, con θ de 0 a 360 grados, están todos sobre el borde de la elipse. seq(0, 360)

genera los ángulos, ones repite b en cada fila, ~ pega dos columnas y el apóstrofo traspone. EP pone en
la última columna la variable del eje horizontal, como pide plot. Como gnuplot escribe los decimales con
punto, set force_decpoint on cambia el separador antes de componer, con sprintf, las dos órdenes de
gnuplot que dibujan el rectángulo de los intervalos y marcan la estimación; dentro del bloque plot (práctica
C de la sesión 11, actividad 2 (html)), @rect y @est insertan esas cadenas, y la última literal marca el
origen.

en línea de comandos:

matrix L = cholesky(V)
matrix th = seq(0, 360)' * $pi/180
matrix E = ones(rows(th), 1)*b' + sqrt(2*Fc) * (L * (cos(th) ~ sin(th))')'
matrix EP = E[,2] ~ E[,1]
set force_decpoint on
string rect = sprintf("set object 1 rect from %g,%g to %g,%g fs empty border lc rgb 'gray40' dt 2", iL, jL, iR, jR)
string est = sprintf("set label 'estimacion' at %g,%g point pt 7 lc rgb 'red' offset 1,1", b[1], b[2])
plot EP

options with-lines fit=none
literal set title "Elipse de confianza al 95% y rectangulo de los dos intervalos"
literal set xlabel "coeficiente de nox"
literal set ylabel "coeficiente de lnox"
literal set key off
literal @rect
literal @est
literal set label "(0,0)" at 0,0 point pt 7 lc rgb "black" offset 1,1

end plot --output="elipse_nox_lnox.png"

Figura 1: La elipse de confianza al 95 % para (βnox, βlnox) y, en trazo discontinuo, el rectángulo de los dos
intervalos al 95 %. El origen está dentro del rectángulo y fuera de la elipse.

Lo que debe observar

El origen está dentro del rectángulo, porque cada intervalo contiene el cero, y fuera de la elipse, porque
el F del par (0, 0) es 28,2 frente a un valor crítico de 3,0. Es la Actividad 5 dibujada: dos t pequeños y un
F grande no se contradicen, porque el rectángulo y la elipse son regiones distintas.

La elipse es una aguja inclinada hacia abajo. La correlación entre los dos estimadores es −0,993, de signo
contrario a la correlación entre los dos regresores, 0,994. Los datos casi no distinguen nox de lnox, y por eso
solo fijan una combinación de los dos coeficientes: aceptan los pares en los que, si uno sube, el otro baja en
la proporción de la pendiente del eje mayor, y rechazan casi todos los demás. Es lo que la lección 14 llamará
“Por qué oscilan los coeficientes”, visto aquí antes como región de confianza: lo que una coordenada gana lo
pierde la otra. La práctica B de esta sesión (actividad 3 (html)) dibuja la misma figura con una correlación
de −0,31: una elipse apenas inclinada, en la que dos esquinas del rectángulo caen dentro y dos fuera.

Preguntas de interpretación para la clase
1. En la Actividad 2, SRCr − SRC vale 4,99 · 109. ¿Qué longitud al cuadrado es esa, en la figura de la

proyección en dos pasos de la lección 13? ¿Por qué no puede ser negativa?

2. El coeficiente de rooms pasa de 7000,6 en el modelo completo a 9119,6 en el restringido de la Actividad
2. ¿Qué ha cambiado, si rooms es el mismo vector?

3. En la Actividad 3 el F y el t2 coinciden. ¿Qué información tiene el t que el F no tiene? ¿Y en qué
situación el t no sirve y el F sí?

4. En la Actividad 4 el valor p es 0,060. Un compañero concluye “queda demostrado que los dos coeficientes
son iguales”. Diga dos razones por las que esa frase es incorrecta.

5. Con la restricción de la Actividad 4, ¿cuánto vale el error estándar del coeficiente común, y por qué es
menor que los errores estándar de nox y stratio en el modelo completo?

6. En la Actividad 6, nox y lnox crecen juntos (correlación 0,994) y la elipse se inclina hacia abajo
(correlación entre los estimadores −0,993). ¿Por qué los signos son opuestos? ¿Qué pares de valores
acepta la elipse y cuáles rechaza?

Para profundizar
Lección 13 (lección 13), en especial “Restricciones lineales: modelos anidados” (con la figura de la
proyección en dos pasos) y “Tres restricciones sobre hprice2”.

Wooldridge, J. M. (2020). Introductory Econometrics, cap. 4, sección 4.5 (contraste de restricciones
lineales múltiples) y el ejemplo 4.9.

Manual de Gretl: comandos restrict y omit; accesores $ess, $df, $test y $pvalue.

Código completo de la práctica

# ------------------------------------------------------------
# Copyright (C) 2026 Marcos Bujosa
# Licencia: GNU General Public License v3.0 o posterior
# Este código es software libre y puede ser redistribuido y/o modificado bajo los términos de la GPL.
# Ver el archivo LICENSE del repositorio para más detalles.
# ------------------------------------------------------------

# Los dos primeros comandos son necesarios para que Gretl guarde los resultados de la práctica en el
directorio de trabajo↪→

# al ejecutar lo siguiente desde un terminal (use los nombres y ruta que correspondan)
#
# DIRECTORIO="Nombre_Directorio_trabajo" gretlcli -b ruta/nombre_fichero_de_la_practica.inp
#
# Si esto no le funciona en su sistema, comente las siguientes dos líneas y sitúese en el directorio de

trabajo de gretl↪→

# que corresponda (configure dicho directorio de trabajo desde la ventana principal de Gretl).

string directory = getenv("DIRECTORIO")
set workdir "@directory"

set verbose off

open hprice2.gdt --quiet
outfile --quiet completo.txt

ols price const rooms nox stratio
scalar n = $nobs
scalar k = 4
scalar SRC = $ess
scalar s2 = SRC/(n-k)
printf "\nSRC = %.4e n-k = %d s^2 = SRC/(n-k) = %.4e\n", SRC, n-k, s2

end outfile

outfile --quiet dosExclusiones.txt
ols price const rooms --quiet
scalar SRCr = $ess
scalar q = 2
scalar F = ((SRCr - SRC)/q)/s2
printf "Modelo restringido (price ~ rooms): SRC_r = %.4e\n", SRCr
printf "Aumento de la SRC: SRC_r - SRC = %.4e\n", SRCr - SRC
printf "F = ((SRC_r - SRC)/%d)/(SRC/(n-k)) = %.3f\n", q, F
printf "F_c(%d,%d) al 5%% = %.3f valor p = %.3g\n", q, n-k, critical(F, q, n-k, 0.05), pvalue(F, q,

n-k, F)↪→

end outfile

outfile --quiet restrictDos.txt
ols price const rooms nox stratio --quiet
restrict

b[nox] = 0
b[stratio] = 0

end restrict
end outfile

outfile --quiet unaExclusion.txt
ols price const rooms nox --quiet
scalar SRCr1 = $ess
scalar F1 = ((SRCr1 - SRC)/1)/s2
ols price const rooms nox stratio --quiet
scalar t_st = $coeff(stratio)/$stderr(stratio)
printf "Modelo restringido (price ~ rooms + nox): SRC_r = %.4e\n", SRCr1
printf "F = (SRC_r - SRC)/(SRC/(n-k)) = %.3f\n", F1
printf "t de stratio en el modelo completo = %.3f ; t^2 = %.3f\n", t_st, t_st^2
printf "F_c(1,%d) al 5%% = %.3f ; t_c^2 = %.3f\n", n-k, critical(F, 1, n-k, 0.05), critical(t, n-k,

0.025)^2↪→

end outfile

outfile --quiet igualdad.txt
series nox_mas_stratio = nox + stratio
ols price const rooms nox_mas_stratio --quiet
scalar SRCr2 = $ess
scalar F2 = ((SRCr2 - SRC)/1)/s2
printf "Modelo restringido (price ~ rooms + (nox + stratio)): SRC_r = %.4e\n", SRCr2
printf "Aumento de la SRC = %.4e\n", SRCr2 - SRC
printf "F = %.3f F_c(1,%d) al 5%% = %.3f valor p = %.4f\n", F2, n-k, critical(F, 1, n-k, 0.05),

pvalue(F, 1, n-k, F2)↪→

end outfile

outfile --quiet restrictIgualdad.txt
ols price const rooms nox stratio --quiet
restrict

b[nox] - b[stratio] = 0
end restrict

end outfile

outfile --quiet apertura.txt
ols price const rooms nox lnox
restrict

b[nox] = 0
b[lnox] = 0

end restrict
end outfile

outfile --quiet elipse.txt
ols price const rooms nox lnox --quiet
scalar tc = critical(t, $df, 0.025)
scalar Fc = critical(F, 2, $df, 0.05)
scalar gl = $df
matrix b = $coeff[3:4]
matrix V = $vcv[3:4,3:4]
scalar corrEst = V[1,2]/sqrt(V[1,1]*V[2,2])
scalar iL = b[1] - tc*sqrt(V[1,1])
scalar iR = b[1] + tc*sqrt(V[1,1])
scalar jL = b[2] - tc*sqrt(V[2,2])
scalar jR = b[2] + tc*sqrt(V[2,2])
restrict --quiet

b[nox] = 0
b[lnox] = 0

end restrict
printf "IC 95%% nox: [%.1f ; %.1f] IC 95%% lnox: [%.1f ; %.1f]\n", iL, iR, jL, jR
printf "Correlacion entre los estimadores de nox y lnox: %.3f\n", corrEst
printf "El par (0,0): F = %.3f F_c(2,%d) = %.3f valor p = %.3g\n", $test, gl, Fc, $pvalue

end outfile

matrix L = cholesky(V)
matrix th = seq(0, 360)' * $pi/180
matrix E = ones(rows(th), 1)*b' + sqrt(2*Fc) * (L * (cos(th) ~ sin(th))')'
matrix EP = E[,2] ~ E[,1]
set force_decpoint on
string rect = sprintf("set object 1 rect from %g,%g to %g,%g fs empty border lc rgb 'gray40' dt 2", iL, jL,

iR, jR)↪→

string est = sprintf("set label 'estimacion' at %g,%g point pt 7 lc rgb 'red' offset 1,1", b[1], b[2])
plot EP

options with-lines fit=none
literal set title "Elipse de confianza al 95% y rectangulo de los dos intervalos"
literal set xlabel "coeficiente de nox"
literal set ylabel "coeficiente de lnox"
literal set key off
literal @rect
literal @est
literal set label "(0,0)" at 0,0 point pt 7 lc rgb "black" offset 1,1

end plot --output="elipse_nox_lnox.png"

Enlace al guión: S20-Prct-A-hprice2-restricciones.inp

Respuestas
1. El aumento de la SRC. Es ∥ŷ−ŷr∥2

e, el cuadrado de la distancia entre el ajuste completo y el restringido:
el cateto que separa las dos proyecciones en la figura de la lección 13. Por Pitágoras anidado, SRCr =
SRC + ∥ŷ − ŷr∥2

e, y un cuadrado no es negativo: el residuo restringido nunca es más corto que el libre,
porque se proyecta sobre un subespacio contenido en el otro.

2. El coeficiente de rooms cambia. El vector rooms es el mismo, pero la coordenada de un vector en una
base depende de toda la base. En el modelo completo, el coeficiente de rooms es la coordenada de ŷ
en la base {1, rooms, nox, stratio}; en el restringido, la de ŷr en la base {1, rooms}. Son dos puntos
distintos descritos en bases distintas. La lección 14 lo dirá con la parte propia de cada regresor: en el
modelo completo, rooms se mide con lo que no comparte con nox y stratio.

3. Qué tiene el t que no tiene el F . El signo, y con él la dirección del efecto: t = −9,06 dice que el coeficiente
estimado de stratio es negativo y lejos de cero; F = 82,1 solo dice que quitarla cuesta mucho. El t
no sirve cuando la hipótesis involucra más de un coeficiente a la vez, como en las Actividades 2 y 5:
dos t pequeños no equivalen a un F pequeño, y la Actividad 5 lo muestra.

4. “Queda demostrado que son iguales”. Primera razón: no rechazar no es demostrar; datos como estos
no serían raros si los dos coeficientes fuesen iguales, y tampoco lo serían con muchas otras diferencias
entre ellos, y p = 0,060 está muy cerca del umbral. Segunda razón: la igualdad que se contrasta es entre
coeficientes de variables medidas en unidades sin relación, así que ni siquiera tiene contenido económico
que demostrar. Un intervalo para βnox − βstratio informaría más que el veredicto.

5. El error estándar del coeficiente común. Es 108,9 en la salida de restrict, frente a 236,3 y 128,8 en
el modelo completo. Al fundir los dos regresores en uno, la parte propia de la suma es más larga que
la de cada sumando: la dirección que nox y stratio compartían deja de restar longitud. Es uno de los
remedios que la lección 14 propondrá para la colinealidad, imponer una restricción que la muestra no
rechaza, visto aquí en acción antes de la lección.

6. Signos opuestos. Si nox y lnox son casi el mismo vector, el ajuste depende casi solo de la suma de sus
dos contribuciones: subir el coeficiente de uno y bajar el del otro en la proporción adecuada deja el
ajuste casi igual, y los datos no distinguen esos pares. Por eso la muestra acepta una franja estrecha
de pares con desvíos de signo contrario y rechaza los desvíos del mismo signo, que sí cambian el ajuste.
Esa franja es la elipse, y su inclinación negativa es la correlación negativa entre los dos estimadores:
en réplicas de la muestra, cuando una coordenada sale alta la otra sale baja, como explicará la lección
14 en “Por qué oscilan los coeficientes” (la práctica A de la sesión 22 lo mostrará con dos mil réplicas
alineadas sobre una recta de pendiente negativa). La elipse contiene el origen solo si la franja pasa por
él, y aquí no pasa: el F rechaza (0, 0) aunque cada intervalo, que es la sombra de la elipse sobre su eje,
contenga el cero.

13


